
> //We verify the assertions made in Section 2 about the existence of a
> //quartic surface Q which is singular at the 15 singular points
> //of the Van der Geer--Zagier quintic V:
> 
> P3<x,y,z,w>:=ProjectiveSpace(Rationals(),3);
> t:=-(x+y+z+w);
> s:=[x^i+y^i+z^i+w^i+t^i:i in [1..5]];
> V:=Surface(P3,12*s[5]-5*s[2]*s[3]);
> pts:=SingularPoints(V);#pts eq 15;
true
> HasSingularPointsOverExtension(V) eq false;
true
> L4:=LinearSystem(P3,4);
> L:=LinearSystem(L4,[P3!x:x in pts],[2:x in pts]);
> #Sections(L) eq 1;
true
> Q:=Surface(P3,4*s[4]-s[2]^2);
> Q eq Surface(P3,Sections(L)[1]);
true
> #SingularPoints(Q) eq 15;
true
> HasSingularPointsOverExtension(Q) eq false;
true
> pc:=PrimeComponents(V meet Q);pc;
[
   Scheme over Rational Field defined by
   x^2 + x*y + y^2 - w^2,
   z + w,
   Scheme over Rational Field defined by
   x^2 + x*z + z^2 - w^2,
   y + w,
   Scheme over Rational Field defined by
   x^2 - z^2 + x*w + w^2,
   y + z,
   Scheme over Rational Field defined by
   x^2 - z^2 - z*w - w^2,
   y + z + w,
   Scheme over Rational Field defined by
   y^2 + y*z + z^2 - w^2,
   x + y + z,
   Scheme over Rational Field defined by
   y^2 + y*z + z^2 - w^2,
   x + w,
   Scheme over Rational Field defined by
   y^2 - z^2 + y*w + w^2,
   x + y + w,
   Scheme over Rational Field defined by



   y^2 - z^2 + y*w + w^2,
   x + z,
   Scheme over Rational Field defined by
   y^2 - z^2 - z*w - w^2,
   x + y,
   Scheme over Rational Field defined by
   y^2 - z^2 - z*w - w^2,
   x + z + w
]
> [#Points(pc[i] meet SingularSubscheme(V)) eq 6:i in [1..10]];
[ true, true, true, true, true, true, true, true, true, true ]
> 
> K<e>:=CyclotomicField(5);
> P<x,y,z,w>:=ProjectiveSpace(K,3);
> 
> //The following sequences of monomials s_4 and s_5 are invariant
> //under the action (x:y:z:w)->(x:e*y:e^2*z:e^3*w).
> //We compute the elements of s_4 which are singular at the point (1:1:1:1).
>  
> s4:=[x^4,y^3*z,x*y*z^2,x*y^2*w,x^2*z*w,z^2*w^2,y*w^3];
> s5:=[x^5,x^3*z*w,x^2*y^2*w,x^2*y*z^2,x*y^3*z,x*y*w^3,x*z^2*w^2,y^5,y^2*z*w^2\
,y*z^3*w,z^5,w^5];
> LinearSystem(LinearSystem(P,s4),P![1,1,1,1],2);
Linear system on Projective Space of dimension 3 over Cyclotomic Field of order 
5 and degree 4
Variables: x, y, z, w
with 3 sections:

x^4 - 2*x^2*z*w + z^2*w^2
y^3*z - 2*x*y^2*w + x^2*z*w - z^2*w^2 + y*w^3
x*y*z^2 - x*y^2*w - z^2*w^2 + y*w^3

> A12<x,y,z,w,a,b,c,d,U,V,N1,N2>:=AffineSpace(K,12);
> s:=[
>     x^4 - 2*x^2*z*w + z^2*w^2,
>     y^3*z - 2*x*y^2*w + x^2*z*w - z^2*w^2 + y*w^3,
>     x*y*z^2 - x*y^2*w - z^2*w^2 + y*w^3
> ];
> F:=s[1]+U*s[2]+V*s[3];
> 
> //We want to compute U, V such that {F=0} has double
> //points at (the orbits of) (1:1:1:1), (x:y:z:w) and (a:b:c:d).
> 
> M:=Submatrix(HessianMatrix(Scheme(A12,F)),1,1,4,4);
> m:=Minors(Evaluate(M,[1,1,1,1,a,b,c,d,U,V,N1,N2]),3)[1];
> 
> //If there is a number N_1 such that 1+N_1*m=0 at (1:1:1:1),
> //then the minor m\not= 0 and the double point at (1:1:1:1) is ordinary.
> //This reduces the probability of getting a non-normal surface.



> 
> Hx:=[F] cat [Derivative(F,i):i in [1..4]];
> Ha:=[Evaluate(Hx[i],[a,b,c,d,a,b,c,d,U,V,N1,N2]):i in [1..#Hx]];
> 
> //To search for the 15 double points, we define the scheme S_C of the points 
> //such that H_x=H_a=0 and m\not= 0.
> 
> SC:=Scheme(A12,Hx cat Ha cat [w-1,d-1,1+N1*m]);
> 
> //We impose extra conditions to ensure that the points
> //(1:1:1:1), (x:y:z:w) and (a:b:c:d) are in different Z_5 orbits:
> 
> SC:=Scheme(SC,[1+N2*(x^5-1)*(x^5-a^5)*(a^5-1)]);
> Dimension(SC) eq 0;
true
> 
> //We compute the points in P^3 corresponding to one of the points of S_C:
> 
> p:=Points(SC)[1];
> p:=[P![1,1,1,1],P![p[1],p[2],p[3],p[4]],P![p[5],p[6],p[7],p[8]]];
> 
> //The unique Z_5 invariant quartic with nodes at (the
> //orbits of) the above three points:
> 
> L:=LinearSystem(LinearSystem(P,s4),p,[2,2,2]);
> Q:=Scheme(P,Sections(L)[1]);
> r:=SingularPoints(Q);
> #r eq 16;
true
> 
> //The quartic Q has 16 nodes.
> //One of these points is (0:0:1:0), a fixed point for the action of Z_5.
> //Now we compute the linear system of invariant quintics with nodes at
> //the 15 points different from (0:0:1:0):
> 
> L:=LinearSystem(LinearSystem(P,s5),p,[2,2,2]);
> #Sections(L) eq 2;
true
> 
> //It remains to find the element of this pencil with 15 cusps.
> 
> A5<X,Y,Z,W,b>:=AffineSpace(K,5);
> h:=hom<CoordinateRing(P)->CoordinateRing(A5)|[X,Y,Z,W]>;
> F:=h(Sections(L)[1])+b*h(Sections(L)[2]);
> M:=HessianMatrix(Scheme(A5,F));RemoveColumn(~M,5);RemoveRow(~M,5);
> H:=[Evaluate(Minors(M,3)[i],[1,1,1,1,b]):i in [1..#Minors(M,3)]];
> 
> //If all these minors vanish, the surface {F=0} has a non-ordinary
> //double point at (1:1:1:1).



> 
> SC:=Scheme(A5,H cat [X,Y,Z,W]);
> Dimension(SC) eq 0;
true
> 
> //The points in S_C give three possibilities for b. One of these corresponds 
> //to a quintic surface with 15 ordinary cusps (A_2 singularities):
> 
> b:=Points(SC)[1][5];
> F:=Sections(L)[1]+b*Sections(L)[2];
> S:=Surface(P,F);
> r:=SingularPoints(S);
> 
> //There are exactly 15 singular points:
> 
> #r eq 15;
true
> HasSingularPointsOverExtension(S) eq false;
true
> 
> //We confirm that these singularities are of type A_2:
> 
> for x in r do IsSimpleSurfaceSingularity(S!x);end for;
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
> 
> //The surfaces S and Q meet at the 10 tropes of Q not
> //containing the point (0:0:1:0):
> 
> pp:=PrimeComponents(Q meet S);
> [#Points(pp[i] meet SingularSubscheme(S)):i in [1..10]];
[ 6, 6, 6, 6, 6, 6, 6, 6, 6, 6 ]
> 
> P<x,y,z,w>:=P;
> psi:=map<P->P|[x,y*e,z*e^2,w*e^3]>;
> 



> //These conics are divided into two Z_5 orbits T_1, T_2:
> 
> T1:=&join[pp[i]:i in [1,2,5,6,9]];
> T1 eq psi(T1);
true
> T2:=&join[pp[i]:i in [3,4,7,8,10]];
> T2 eq psi(T2);
true
> [Multiplicity(T1,r[i]):i in [4,5,6]] eq [1,2,3];
true
> [Multiplicity(T2,r[i]):i in [4,5,6]] eq [3,2,1];
true
> 
> //We define T_3 as the Z_5 orbit of plane quintics corresponding
> //to 5 of the 6 tropes of Q through the fixed point (0:0:1:0):
> 
> L1:=LinearSystem(LinearSystem(P,1),[P!r[i]:i in [1,3,11,12,13]]);
> t:=[Scheme(P,Sections(L1)[1])];
> #Points(t[1] meet SingularSubscheme(S)) eq 5;
true
> for i in [1..4] do t:=t cat [psi(t[#t])];end for;
> T3:=(&join t) meet S;
> T3 eq psi(T3);
true
> [Multiplicity(T3,r[i]):i in [4,5,6]] eq [4,2,4];
true
> 
> //The strict transform of T_3 does not intersect the strict transform
> //of T_1+T_2:
> 
> Points(T3 meet (T1 join T2)) eq r;
true
> 
> //The resolution of each cusp is an union of two (-2)-curves A_i,A_i'.
> //We compute the intersection number of T_1, T_2 and T_3 with these curves.
> //First we blowup the quintic at a cusp:
> 
> T:=[T1,T2,T3];
> for j in [4,5,6] do
for>  X,mp:=Blowup(S,P!r[j]);
for> 
for> //We define the exceptional divisor E and the (-2)-curves A_i,A_i':
for>  E:=(P!r[j]) @@ mp;
for>  A:=PrimeComponents(E meet Complement(X,E));
for> 
for> //and compute the intersection number of the strict transform of
for> //T_i with the (-2)-curves:
for> 
for>  for i in [1,2,3] do



for|for>   t:=Complement(Scheme(S,DefiningEquations(T[i])) @@ mp,E);
for|for>   #Points(t meet A[1]),#Points(t meet A[2]);
for|for>  end for;
for> end for;
1 0
1 2
2 2
0 2
2 0
1 1
1 2
1 0
2 2
> 
> //This shows that the intersection matrix of the curves, in the Godeaux
> //surface, corresponding to the A_2 configurations of the three cusps and
> //the orbits T_1, T_2 and T_3 is:
> 
> M:=SymmetricMatrix([-2,1,-2,0,0,-2,0,0,1,-2,0,0,0,0,-2,0,0,0,0,1,-2,\
> 1,0,0,2,1,2,-4,1,2,2,0,1,0,0,-4,2,2,1,1,2,2,0,0,-1]);
> M;
[-2  1  0  0  0  0  1  1  2]
[ 1 -2  0  0  0  0  0  2  2]
[ 0  0 -2  1  0  0  0  2  1]
[ 0  0  1 -2  0  0  2  0  1]
[ 0  0  0  0 -2  1  1  1  2]
[ 0  0  0  0  1 -2  2  0  2]
[ 1  0  0  2  1  2 -4  0  0]
[ 1  2  2  0  1  0  0 -4  0]
[ 2  2  1  1  2  2  0  0 -1]
> Determinant(M) eq 0;
true
> Nullspace(M);
RSpace of degree 9, dimension 1 over Integer Ring
Echelonized basis:
( 2  4  2 -2 -2 -4 -3  3  0)
> 
> //We define the Segre cubic S_3, the Igusa quartic I_4
> //and compute the singular set of I_4:
> 
> P4<x,y,z,w,t>:=ProjectiveSpace(Rationals(),4);
> S3:=Scheme(P4,x^3+y^3+z^3+w^3+t^3+(-x-y-z-w-t)^3);
> rho:=map<P4->P4|Basis(JacobianIdeal(DefiningEquation(S3)))>;
> I4:=rho(S3);
> SI4:=SingularSubscheme(I4);
> 
> //The S_5 symmetric quintics are generated by:
> 
> e:=[ElementarySymmetricPolynomial(CoordinateRing(P4),i):i in [1..5]];



> s5:=[e[1]^5,e[1]^3*e[2],e[1]^2*e[3],e[1]*e[4],e[1]*e[2]^2,e[2]*e[3],e[5]];
> 
> //We fix 15 singular points of I_4
> //and compute the symmetric quintics singular at these points:
> 
> r:=Points(Scheme(SI4,x+2*y+3*z+4*w+5*t));                                   \
      
> L5:=LinearSystem(P4,s5);
> L:=LinearSystem(L5,[P4!x:x in r],[2:x in r]);
> 
> //A generic element of the pencil L is singular exactly at
> //the 15 singular lines of the Igusa:
> 
> X1:=SingularSubscheme(Scheme(P4,Sections(L)[1]));
> X2:=SingularSubscheme(Scheme(P4,Sections(L)[2]));
> SI4 eq ReducedSubscheme(X1);
true
> SI4 eq ReducedSubscheme(X2);
true
> 
> //We compute the element of L which contains 15 cuspidal lines:
> 
> A<X,Y,Z,W,T,b>:=AffineSpace(Rationals(),6);
> h:=hom<CoordinateRing(P4)->CoordinateRing(A)|[X,Y,Z,W,T]>;
> F:=h(Sections(L)[1])+b*h(Sections(L)[2]);
> 
> //We compute b such that the quintic {F = 0} has cusps
> //at the 15 points above: 
> 
> H:=HessianMatrix(Scheme(A,F));RemoveColumn(~H,6);RemoveRow(~H,6);
> G:=[F] cat [Derivative(F,i):i in [1..5]] cat Minors(H,3);
> 
> //The vanishing of these minors implies non-ordinary double points. 
> 
> G:=[Evaluate(G[j],Coordinates(r[i]) cat [b]):j in [1..#G],i in [1..#r]];
> S:=Scheme(A,G cat [X,Y,Z,W,T]);
> Dimension(S) eq 0;
true
> PointsOverSplittingField(S);
{@ (0, 0, 0, 0, 0, -5/7) @}
Algebraically closed field with no variables over Rational Field
> 
> //This gives b = -5/7.
> 
> //The quintic threefold with 15 cuspidal lines:
> 
> F:=Sections(L)[1]-5/7*Sections(L)[2];
> Q:=Scheme(P4,F);
> SI4 eq ReducedSubscheme(SingularSubscheme(Q));



true
> Degree(SingularSubscheme(Q)) eq 30;
true
> F;
x^5 - 5/7*x^4*y - 2/7*x^3*y^2 - 2/7*x^2*y^3 - 5/7*x*y^4 + y^5 - 5/7*x^4*z + 
   4/7*x^3*y*z + 2/7*x^2*y^2*z + 4/7*x*y^3*z - 5/7*y^4*z - 2/7*x^3*z^2 + 
   2/7*x^2*y*z^2 + 2/7*x*y^2*z^2 - 2/7*y^3*z^2 - 2/7*x^2*z^3 + 4/7*x*y*z^3 - 
   2/7*y^2*z^3 - 5/7*x*z^4 - 5/7*y*z^4 + z^5 - 5/7*x^4*w + 4/7*x^3*y*w + 
   2/7*x^2*y^2*w + 4/7*x*y^3*w - 5/7*y^4*w + 4/7*x^3*z*w - 4/7*x^2*y*z*w - 
   4/7*x*y^2*z*w + 4/7*y^3*z*w + 2/7*x^2*z^2*w - 4/7*x*y*z^2*w + 2/7*y^2*z^2*w 
   + 4/7*x*z^3*w + 4/7*y*z^3*w - 5/7*z^4*w - 2/7*x^3*w^2 + 2/7*x^2*y*w^2 + 
   2/7*x*y^2*w^2 - 2/7*y^3*w^2 + 2/7*x^2*z*w^2 - 4/7*x*y*z*w^2 + 2/7*y^2*z*w^2 
   + 2/7*x*z^2*w^2 + 2/7*y*z^2*w^2 - 2/7*z^3*w^2 - 2/7*x^2*w^3 + 4/7*x*y*w^3 - 
   2/7*y^2*w^3 + 4/7*x*z*w^3 + 4/7*y*z*w^3 - 2/7*z^2*w^3 - 5/7*x*w^4 - 
   5/7*y*w^4 - 5/7*z*w^4 + w^5 - 5/7*x^4*t + 4/7*x^3*y*t + 2/7*x^2*y^2*t + 
   4/7*x*y^3*t - 5/7*y^4*t + 4/7*x^3*z*t - 4/7*x^2*y*z*t - 4/7*x*y^2*z*t + 
   4/7*y^3*z*t + 2/7*x^2*z^2*t - 4/7*x*y*z^2*t + 2/7*y^2*z^2*t + 4/7*x*z^3*t + 
   4/7*y*z^3*t - 5/7*z^4*t + 4/7*x^3*w*t - 4/7*x^2*y*w*t - 4/7*x*y^2*w*t + 
   4/7*y^3*w*t - 4/7*x^2*z*w*t + 8/7*x*y*z*w*t - 4/7*y^2*z*w*t - 4/7*x*z^2*w*t 
   - 4/7*y*z^2*w*t + 4/7*z^3*w*t + 2/7*x^2*w^2*t - 4/7*x*y*w^2*t + 
   2/7*y^2*w^2*t - 4/7*x*z*w^2*t - 4/7*y*z*w^2*t + 2/7*z^2*w^2*t + 4/7*x*w^3*t 
   + 4/7*y*w^3*t + 4/7*z*w^3*t - 5/7*w^4*t - 2/7*x^3*t^2 + 2/7*x^2*y*t^2 + 
   2/7*x*y^2*t^2 - 2/7*y^3*t^2 + 2/7*x^2*z*t^2 - 4/7*x*y*z*t^2 + 2/7*y^2*z*t^2 
   + 2/7*x*z^2*t^2 + 2/7*y*z^2*t^2 - 2/7*z^3*t^2 + 2/7*x^2*w*t^2 - 
   4/7*x*y*w*t^2 + 2/7*y^2*w*t^2 - 4/7*x*z*w*t^2 - 4/7*y*z*w*t^2 + 
   2/7*z^2*w*t^2 + 2/7*x*w^2*t^2 + 2/7*y*w^2*t^2 + 2/7*z*w^2*t^2 - 2/7*w^3*t^2 
   - 2/7*x^2*t^3 + 4/7*x*y*t^3 - 2/7*y^2*t^3 + 4/7*x*z*t^3 + 4/7*y*z*t^3 - 
   2/7*z^2*t^3 + 4/7*x*w*t^3 + 4/7*y*w*t^3 + 4/7*z*w*t^3 - 2/7*w^2*t^3 - 
   5/7*x*t^4 - 5/7*y*t^4 - 5/7*z*t^4 - 5/7*w*t^4 + t^5
> 
> //The coefficients of F in Sections(L5):
> 
> CoefficientMap(L5)(F);
(   1 -5/7 -2/7  4/7  2/7 -4/7  8/7)
> 
> //We verify that a random hyperplane section is a quintic surface
> //with 15 cusps:
> 
> a:=[Random(1,100):i in [1..5]];
> S:=Surface(P4,[F,a[1]*x+a[2]*y+a[3]*z+a[4]*w+a[5]*t]);
> #SingularPoints(S) eq 15;
true
> r:=SingularPoints(S);
> for x in r do IsSimpleSurfaceSingularity(S!x);end for;
true A 2
true A 2
true A 2
true A 2
true A 2



true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
> 
> 
> //Now we compute an hyperplane section of the above quintic threefold
> //which is a quintic surface with 17 cusps:
> 
> A<X,Y,Z,W,T,a,b,c,d>:=AffineSpace(Rationals(),9);
> h:=hom<CoordinateRing(P4)->CoordinateRing(A)|[X,Y,Z,W,T]>;
> G:=a*X+b*Y+c*Z+d*W+T;
> f:=Evaluate(h(F),T,-(a*X+b*Y+c*Z+d*W));
> 
> //So {f=0} is an hyperplane section of the quintic threefold.
> 
> H:=HessianMatrix(Scheme(A,f));
> H:=Submatrix(H,1,1,4,4);
> min1:=Minors(H,3);
> 
> //If all these minors vanish, then the double point is non-ordinary.
>  
> J:=[JacobianSequence(h(F))[i]:i in [1..5]];
> min2:=Minors(Matrix([J,[a,b,c,d,1]]),2);
> 
> //If all these minors vanish, then the hyperplane is tangent to the quintic. 
> 
> H:=[T-1,h(F),G] cat min1 cat min2;
> SC:=Scheme(A,H cat [X,Y-2]);
> pp:=PrimeComponents(SC);
> [Dimension(x):x in pp];
[ 3, 3, 0, 0, 0 ]
> 
> //Since S_C has zero dimensional components, we can compute some points.
> //These data allow us to compute the quintic surface with 17 cusps:
> 
> p:=Points(pp[4]);
> a:=p[1][6];b:=p[1][7];c:=p[1][8];d:=p[1][9];
> T:=a*x+b*y+c*z+d*w+t;T;
3/7*x - 13/7*y - 13/7*z + w + t
> Q:=Surface(P4,[F,T]);
> r:=Points(SingularSubscheme(Q));
> #r;



17
> for i in [1..#r] do IsSimpleSurfaceSingularity(Q!r[i]);end for;
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
> HasSingularPointsOverExtension(Q) eq false;
true
> 
> 
> 
> 
> //Finally we compute hyperplane sections of the quintic threefold which
> //are quintic surfaces with the following singular sets:
> //16A_2, 15A_2 + A_3, 15A_2 + D_4.
> 
> 
> //The cases 16A_2 and 15A_2 + A_3:
> 
> S:=Scheme(A,H cat [X,Y-1]);
> pp:=PrimeComponents(S);
> [Dimension(x):x in pp];
[ 4, 0, 0 ]
> 
> p,K:=PointsOverSplittingField(pp[2]);
> K;
Algebraically closed field with 6 variables over Rational Field
Defining relations:
[
   r6^6 + 10*r6^5 + 85/2*r6^4 + 8*r6^3 - 199/16*r6^2 + 531/8*r6 + 857/16,
   r5^6 + 10*r5^5 + 85/2*r5^4 + 8*r5^3 - 199/16*r5^2 + 531/8*r5 + 857/16,
   r4^6 + 10*r4^5 + 85/2*r4^4 + 8*r4^3 - 199/16*r4^2 + 531/8*r4 + 857/16,
   r3^6 + 10*r3^5 + 85/2*r3^4 + 8*r3^3 - 199/16*r3^2 + 531/8*r3 + 857/16,
   r2^6 + 10*r2^5 + 85/2*r2^4 + 8*r2^3 - 199/16*r2^2 + 531/8*r2 + 857/16,
   r1^6 + 10*r1^5 + 85/2*r1^4 + 8*r1^3 - 199/16*r1^2 + 531/8*r1 + 857/16
]



> p,K:=PointsOverSplittingField(pp[3]);
> K;
Algebraically closed field with 2 variables over Rational Field
Defining relations:
[
   r2^2 + 3/2*r2 + 9/64,
   r1^2 + 3/2*r1 + 9/64
]
> 
> R<r1>:=PolynomialRing(Rationals());
> f:=r1^6 + 10*r1^5 + 85/2*r1^4 + 8*r1^3 - 199/16*r1^2 + 531/8*r1 + 857/16;
> K<r1>:=ext<Rationals()|f>;
> R<r2>:=PolynomialRing(K);
> f:=r2^2 + 3/2*r2 + 9/64;
> K<r2>:=ext<K|f>;
> 
> for i in [2,3] do
for>  ppi:=BaseExtend(pp[i],K);
for>  p:=Points(ppi);
for>  Q:=BaseExtend(Scheme(P4,F),K);
for>  PK<x,y,z,w,t>:=Ambient(Q);
for>  a:=p[1][6];b:=p[1][7];c:=p[1][8];d:=p[1][9];
for>  T:=a*x+b*y+c*z+d*w+t;T;
for>  S:=Surface(PK,[DefiningEquation(Q),T]:Check:=false);
for>  r:=Points(SingularSubscheme(S));
for>  #r;
for>  for i in [1..#r] do IsSimpleSurfaceSingularity(S!r[i]);end for;
for>  HasSingularPointsOverExtension(S);
for> end for;
1/4940741*(591952*r1^5 + 5333076*r1^4 + 19905356*r1^3 - 14919889*r1^2 + 
   6126918*r1 + 30787260)*x + y + 1/4940741*(-2367808*r1^5 - 21332304*r1^4 - 
   79621424*r1^3 + 59679556*r1^2 - 29448413*r1 - 133030522)*z + r1*w + t
16
true A 2
true A 3
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2



false
-1/8*x + y + r2*z + (-r2 - 3/2)*w + t
16
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
false
> 
> 
> //The case 15A_2 + D_4:
> 
> S:=Scheme(A,H cat [X-1,Y-1]);
> pp:=PrimeComponents(S);
> [Dimension(x):x in pp];
[ 3, 3, 1, 0 ]
> p,K:=PointsOverSplittingField(pp[4]);
> K;
Algebraically closed field with 4 variables over Rational Field
Defining relations:
[
   r4^4 + 15*r4^3 + 286/3*r4^2 + 155*r4 + 75,
   r3^4 + 15*r3^3 + 286/3*r3^2 + 155*r3 + 75,
   r2^4 + 15*r2^3 + 286/3*r2^2 + 155*r2 + 75,
   r1^4 + 15*r1^3 + 286/3*r1^2 + 155*r1 + 75
]
> 
> R<r1>:=PolynomialRing(Rationals());
> f:=r1^4 + 15*r1^3 + 286/3*r1^2 + 155*r1 + 75;
> K<r1>:=ext<Rationals()|f>;
> 
> pp4:=BaseExtend(pp[4],K);
> p:=Points(pp4);
> Q:=BaseExtend(Scheme(P4,F),K);
> PK<x,y,z,w,t>:=Ambient(Q);
> a:=p[1][6];b:=p[1][7];c:=p[1][8];d:=p[1][9];
> T:=a*x+b*y+c*z+d*w+t;T;



x + y + 1/1440*(-867*r1^3 - 12240*r1^2 - 71854*r1 - 72255)*z + 1/117*(75*r1^3 + 
   1035*r1^2 + 5908*r1 + 4395)*w + t
> S:=Surface(PK,[DefiningEquation(Q),T]:Check:=false);
> r:=Points(SingularSubscheme(S));
> #r;
16
> for i in [1..#r] do IsSimpleSurfaceSingularity(S!r[i]);end for;
true A 2
true A 2
true A 2
true A 2
true D 4
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
true A 2
> HasSingularPointsOverExtension(S) eq false;
true
> 


